# Math 132
# Fall 2007 Exam II

22 October 2007

## Integral Formula:

$$\int \sec(t)^3\,dt = \frac{1}{2}\sec(t)\tan(t) + \frac{1}{2}\ln(|\sec(t)+\tan(t)|) + C$$

**1. Suppose that** $f(x) = 2^{\sqrt{x}}$**. Calculate** $D(f)(9)$**, the derivative of** $f(x)$ **evaluated at** $x = 9$**.)**

a) $\frac{2}{3}\ln(2)$ b) $\frac{2}{9}\ln(2)$ c) $\frac{4}{3}\ln(2)$ d) $\frac{4}{9}\ln(2)$ e) $\frac{4}{27}\ln(2)$

f) $\frac{2}{3\ln(2)}$ g) $\frac{2}{9\ln(2)}$ h) $\frac{4}{3\ln(2)}$ i) $\frac{4}{9\ln(20)}$ j) $\frac{4}{27\ln(2)}$

**Solution ( c )**

```
> f := x -> 2^(sqrt(x));
```

$$f := x \to 2^{\sqrt{x}}$$

```
> D(f)(x);
```

$$\frac{1}{2}\frac{2^{(\sqrt{x})}\ln(2)}{\sqrt{x}}$$

```
> D(f)(9);
```

$$\frac{4}{3}\ln(2)$$

**2. Calculate** $\int_1^3 \log_3(x)\,dx$**.**

a) $2 - \frac{1}{\ln(3)}$ b) $2 + \frac{1}{\ln(3)}$ c) $2 - \ln(3)$ d) $2 + \ln(3)$ e) $2\ln(3) - 1$

f) $2\ln(3)+1$ g) $3-\frac{2}{\ln(3)}$ h) $3+\frac{2}{\ln(3)}$ i) $1-\frac{2}{\ln(3)}$ j) $1+\frac{2}{\ln(3)}$

**Solution ( g )**

```
> Int('log[3](x)',x)= Int(log[3](x),x);
```

$$\int \log_3(x)\,dx = \int \frac{\ln(x)}{\ln(3)}\,dx$$

```
> Int(ln(x)/ln(3),x)=student[intparts](Int(ln(x)/ln(3),x),ln(x));
  #Integration by parts with u=ln(x)
```

$$\int \frac{\ln(x)}{\ln(3)}\,dx = \frac{x\ln(x)}{\ln(3)} - \int \frac{1}{\ln(3)}\,dx$$

```
> Int(ln(x)/ln(3),x)=
  value(student[intparts](Int(ln(x)/ln(3),x),ln(x)));
```

$$\int \frac{\ln(x)}{\ln(3)}\,dx = \frac{x\ln(x)}{\ln(3)} - \frac{x}{\ln(3)}$$

```
> Answer := Int('log[3](x)',x = 1..3) = subs(x=3,
  1/ln(3)*x*ln(x)-1/ln(3)*x) -
  subs(x=1,1/ln(3)*x*ln(x)-1/ln(3)*x);
```

$$Answer := \int_1^3 \log_3(x)\,dx = 3 - \frac{2}{\ln(3)} - \frac{\ln(1)}{\ln(3)}$$

```
> Answer := value(rhs(Answer));
```

$$Answer := 3 - \frac{2}{\ln(3)}$$

**3. Suppose that $f(x) = x^{\sqrt{x}}$. Calculate D(f)(4). (D(f)(4) is the derivative of f(x) evaluated at x = 4.)**

a) $1+\ln(2)$ b) $2(2+\ln(2))$ c) $4(2+\ln(2))$ d) $2(1+\ln(2))$ e) $8+\ln(2)$

f) $2+\ln(2)$ g) $4+\ln(2)$ h) $4(1+\ln(2))$ i) $2(4+\ln(2))$ j) $8(1+\ln(2))$

## Solution ( j )

```
> f := x -> x^sqrt(x);
```

$$f := x \to x^{\sqrt{x}}$$

```
> D(f)(x);
```

$$x^{(\sqrt{x})}\left(\frac{1}{2}\frac{\ln(x)}{\sqrt{x}} + \frac{1}{\sqrt{x}}\right)$$

```
> D(f)(4);
```

$$4\ln(4) + 8$$

**4. A radioactive substance has mass 120g at time $t = 4$ and mass 90g at time $t = 6$. What is the mass at $t = 12$ ?**

a) $\frac{1979}{32}$ b) $\frac{1985}{32}$ c) $\frac{1991}{32}$ d) $\frac{1997}{32}$ e) $\frac{1203}{21}$

f) $\frac{1209}{32}$ g) $\frac{1215}{32}$ h) $\frac{1221}{32}$ i) $\frac{1227}{32}$ j) $\frac{1233}{32}$

## Solution ( g )

```
> m:= t -> A*exp(-lambda*t);
```

$$m := t \to A\,e^{(-\lambda t)}$$

```
> eqn1 := m(4) = 120;
```

$$eqn1 := A\,e^{(-4\lambda)} = 120$$

```
> eqn2 := m(6) = 90;
```

$$eqn2 := A\,e^{(-6\lambda)} = 90$$

```
> eqn3 := lhs(eqn1)/lhs(eqn2) = rhs(eqn1)/rhs(eqn2);
```

$$eqn3 := \frac{e^{(-4\lambda)}}{e^{(-6\lambda)}} = \frac{4}{3}$$

```
> eqn4 := combine(lhs(eqn3), exp) = rhs(eqn3);
```

$$eqn4 := e^{(2\lambda)} = \frac{4}{3}$$

```
> eqn5 := lambda = solve(eqn4, lambda);
```

$$eqn5 := \lambda = \frac{1}{2}\ln\left(\frac{4}{3}\right)$$

```
> eqn6 := subs(eqn5, eqn1);  #Substitute this value of lambda
```